

APPENDICE AL CAPITOLO OTTAVO

1) *Un programma per la verifica a pressoflessione di sezioni miste non reagenti a trazione.*

Il programma è relativo (fig. 1) ad una sezione costituita da materiale non reagente a trazione, eventualmente rinforzata con aree concentrate reagenti invece anche a trazione; è quest'ultimo il caso del conglomerato armato, ed n è il rapporto di amplificazione ($n = E_f/E_c$).

Moltiplicando le aree concentrate per tale rapporto, ci si riconduce alla trattazione del D.S. Venant, Se infatti le sezioni restano piane, per calcolare, su un tronco elementare soggetto a $\Delta\varphi$ e Δw , le caratteristiche M ed N , attraverso le ϵ e quindi le σ , occorre moltiplicare le ϵ del conglomerato per E_c e quelle dell'acciaio per E_f , e ciò equivale a considerare aree di acciaio esaltate per E_f/E_c , e moltiplicare anche le ϵ dell'acciaio per E_c . Così infatti ogni area di acciaio è associata alla stessa forza, perchè la σ è ridotta nel rapporto E_c/E_f , ma l'area è esaltata nel rapporto inverso. Ci si riconduce così alla trattazione generale, ed in particolare al calcolo ordinario delle caratteristiche di inerzia.

La sezione è monoconnessa, quindi la frontiera è un'unica linea chiusa; essa si suppone poligonale, ed i vertici si numerano percorrendola in modo da vedere la sezione sempre alla propria sinistra. Il caso della sezione pluriconnessa si riconduce banalmente a quello della sezione monoconnessa, con opportuni tagli; se ne darà un esempio (prob. 2). La curva di frontiera può essere concava o convessa. Il punto C , di coordinate x_0 e y_0 , è il centro di sollecitazione, lo sforzo normale N è negativo (compressione) (in presenza di barre di acciaio, non è considerato il caso, pur possibile, di $N > 0$).

Per le aree metalliche si calcolano una volta per sempre i valori delle caratteristiche d'inerzia rispetto ai due assi x^* ed y^* paralleli per C ed equiversi ad x ed y ; essi sono

$$(S1) \quad S_{x^*} = n \sum_i m_i (y_i - y_0)$$

$$(S2) \quad S_{y^*} = -n \sum_i m_i (x_i - x_0)$$

$$(I1) \quad I_{x^*} = n \sum_i m_i (y_i - y_0)^2 \quad (1)$$

$$(I2) \quad I_{y^*} = n \sum_i m_i (x_i - x_0)^2$$

$$(J1) \quad I_{x^*y^*} = -n \sum_i m_i (y_i - y_0) (x_i - x_0) .$$




Figura 1

Per una coppia di assi a e b ruotata di φ rispetto alla x^*y^* (φ è l'angolo F della fig. 1) si ha

$$(S3) \quad S_a = S_{y^*} \sin \varphi + S_{x^*} \cos \varphi$$

$$(S4) \quad S_b = S_{y^*} \cos \varphi - S_{x^*} \sin \varphi$$

$$(I3) \quad I_a = I_{y^*} \sin^2 \varphi + I_{x^*} \cos^2 \varphi + 2I_{x^*y^*} \sin \varphi \cos \varphi$$

$$(J3) \quad I_{ab} = (I_{x^*} + I_{y^*}) \sin \varphi \cos \varphi + I_{x^*y^*} (\cos^2 \varphi - \sin^2 \varphi).$$

Le distanze del generico vertice dai due assi a e b sono

$$d_{ai} = b_i = - (x_i - x_0) \sin \varphi + (y_i - y_0) \cos \varphi \quad (3)$$

$$d_{bi} = - a_i = - (x_i - x_0) \cos \varphi - (y_i - y_0) \sin \varphi .$$

Il calcolo si esegue, come ben noto, per tentativi. Fissato φ , si cerca (fig. 2) fra tutte le parallele ad a quella, n , il cui centro C^* rispetto alle




Figura 2

aree dei ferri ed all'area tratteggiata della sezione giaccia sulla retta a . Ad ogni angolo φ corrisponde una retta n ed un segmento orientato CC' su a ; il valore di φ che interessa è quello per cui $CC' = 0$.

La ricerca di n per un dato φ si esegue a sua volta, in questo programma, per tentativi; e cioè, data (fig. 3) la generica a' parallela ad a , si cal-




Figura 3

colano di seguito, in relazione alle aree di tutti i ferri ed alla sola parte di sezione sotto a' (che si suppone compressa, e quindi la più vicina a C)

$(A6)$	A	$(I6)$	I_a	
$(S6)$	S_a	$(J6)$	I_{ab}	(4)
$(S7)$	S_b			

e quindi

$$(M1) \quad h_1 = d_{Ga} = \frac{S_a}{A}$$

$$(I8) \quad I_{a'} = I_a - A h_1^2 + A (b_0 - h_1)^2 = \\ = I_a + A (b_0^2 - 2 b_0 h_1)$$

dove b_0 (BO) è la distanza di a' da a , positiva se a' è al disopra di a .
E' poi

$$(S8) \quad S_{a'} = - A (b_0 - h_1)$$

$$(B1) \quad b_1 = d_{Ca'} = \frac{I_{a'}}{S_{a'}}$$

$$(B3) \quad b_3 = d_{Ca} = b_0 + b_1 .$$

Se $b_3 = 0$, a' è la retta parallela ad a per cui C' giace su a , e quindi, è n . La distanza di C' da C è poi data da

$$(C) \quad c = \frac{I_{ab} - S_b b_0}{S_a - A b_0} ; \quad (5)$$

ed infatti è

$$c = \frac{I_{a'b}}{S_{a'}}$$

e, per i teoremi del trasporto,

$$I_{a'b} = I_{ab} - S_b b_0 \\ S_{a'} = S_a - A b_0 .$$

Nella fig. 4 è riportato il listing del programma; è omessa la sola parte riguardante il grafico.

```

70 REM Programma "GLORIA"
150 DEG
160 READ N,N0,X0,Y0
170 READ E,K1,K
180 READ F,H,V,A
190 OPTION BASE 1
200 DIM V(17,2)
210 DIM A(10,3)
220 DIM D(17,2)
230 DIM B(10,3)
240 FOR I=1 TO V
250 FOR J=1 TO 2
260 READ V(I,J)
270 NEXT J
280 NEXT I
290 FOR I=1 TO A
300 FOR J=1 TO 3
310 READ A(I,J)
320 NEXT J
330 NEXT I
340 A1=0
350 S1=0
360 I1=0
370 S2=0
380 I2=0
390 J1=0
400 C0=0
410 FOR I=1 TO A
420 A1=A1+A(I,3)
430 S1=S1+A(I,3)*(A(I,2)-Y0)
440 S2=S2-A(I,3)*(A(I,1)-X0)
450 I1=I1+A(I,3)*(A(I,2)-Y0)^2
460 I2=I2+A(I,3)*(A(I,1)-X0)^2
470 J1=J1-A(I,3)*(A(I,1)-X0)*(A(I,2)-Y0)
480 NEXT I
490 A1=N0*A1
500 S3=N0*S2*SIN(F)+N0*S1*COS(F)
510 S4=N0*S2*COS(F)-N0*S1*SIN(F)
520 I3=N0*I2*SIN(F)^2+N0*I1*COS(F)^2+N0*2*J1*SIN(F)*COS(F)
530 J3=-N0*(I1+I2)*SIN(F)*COS(F)+N0*J1*(COS(F)^2-SIN(F)^2)
540 FOR I=1 TO V
550 D(I,1)=- (V(I,1)-X0)*SIN(F)+(V(I,2)-Y0)*COS(F)
560 D(I,2)=- (V(I,1)-X0)*COS(F)-(V(I,2)-Y0)*SIN(F)
570 NEXT I
580 T1=D(1,1)
590 T2=D(1,1)
600 FOR I=2 TO V
610 IF T1<=D(I,1) THEN 630
620 T1=D(I,1)
630 IF T2>=D(I,1) THEN 650
640 T2=D(I,1)
650 NEXT I
660 M0=10
670 B9=0
680 M=10
690 P0=(T2-T1)/M0

```

Figura 4a

```

700 T0=1
710 B4=0
720 B5=0
730 IF M=10 THEN 760
740 B0=B8+P0*T0
750 GOTO 770
760 B0=T1+P0*T0
770 W=0
780 FOR I=1 TO V-1
790 IF D(I+1,1)-D(I,1)=0 THEN 1060
800 IF D(I+1,1)>=B0 AND D(I,1)>=B0 THEN 1060
810 P=(D(I+1,1)-D(I,1))/M.
820 T=1
830 IF W=1 THEN 890
840 A6=A1
850 S6=S3
860 S7=S4
870 I6=I3
880 J6=J3
890 D9=D(I,2)
900 B=D(I,1)+T*P
910 D=D(I,2)+(D(I+1,2)-D(I,2))/(D(I+1,1)-D(I,1))*(B-D(I,1))
920 IF P<0 THEN 950
930 IF B>B0 THEN 1020
940 GOTO 960
950 IF B-P>B0 THEN 1020
960 D8=(D+D9)/2
970 A6=A6-P*D8
980 S6=S6-P*D8*(B-P/2)
990 S7=S7-P*D8*D8/2
1000 I6=I6-D8*P^3/12-D8*P*(B-P/2)^2
1010 J6=J6-D8*P*(B-P/2)*D8/2
1020 D9=D
1030 T=T+1
1040 IF T<M+1 THEN 900
1050 W=1
1060 NEXT I
1070 H1=S6/A6
1080 I8=I6+A6*(B0^2-2*B0*H1)
1090 S8=A6*(H1-B0)
1100 B1=I8/S8
1110 B3=B0+B1
1120 IF B3*B4>0 THEN 1140
1130 IF B5<0 THEN 1180
1140 B4=B3
1150 B5=B1
1160 T0=T0+1
1170 IF T0<M0+1 THEN 730
1180 IF ABS((B0-B9)/B0)<=E THEN 1280
1190 IF B0<=T2 THEN 1230
1200 M0=2*M0
1210 B8=T1
1220 GOTO 670
1230 B8=B0-2*P0
1240 M=M*2

```

Figura 4b

```

1250 M0=M0*2
1260 B9=B0
1270 GOTO 690
1280 C=(J6-S7*B0)/(S6-A6*B0)
1290 IF ABS(C)<T2/1000 THEN 1550
1300 IF C0=0 THEN 1380
1310 IF C*C0<0 THEN 1460
1320 IF ABS(C)<ABS(C0) THEN 1380
1330 PRINT
1340 PRINT
1350 PRINT "Diminuire val. F"
1360 BEEP
1370 PAUSE
1380 PRINT "Errore C      = ";C
1390 PRINT "Incl.n su x  = ";F
1400 PRINT "Dist.C da n   = ";B0
1410 PRINT
1420 PRINT
1430 F=F+K
1440 C0=C
1450 GOTO 500
1460 F=F-K
1470 K=K/10
1480 C0=0
1490 PRINT "Success.val.F = ";F
1500 PRINT "Success.val.K = ";K
1510 PRINT
1520 PRINT
1530 IF K<K1 THEN 1550
1540 GOTO 500
1550 FOR I=1 TO A
1560 B(I,1)=-((A(I,1)-X0)*SIN(F)+
            (A(I,2)-Y0)*COS(F)
1570 B(I,2)=-((A(I,1)-X0)*COS(F)-
            (A(I,2)-Y0)*SIN(F)
1580 B(I,3)=A(I,3)
1590 NEXT I
1600 Q=0
1610 FOR I=1 TO A-1
1620 IF B(I,1)<=B(I+1,1) THEN 17
            30
1630 V1=B(I,1)
1640 V2=B(I,2)
1650 V3=B(I,3)
1660 B(I,1)=B(I+1,1)
1670 B(I,2)=B(I+1,2)
1680 B(I,3)=B(I+1,3)
1690 B(I+1,1)=V1
1700 B(I+1,2)=V2
1710 B(I+1,3)=V3
1720 Q=0
1730 NEXT I
1740 IF Q=1 THEN 1600
1750 Z=N/S8*(T1-B0)
1760 IF A=0 THEN 1780
1770 Z1=N0*N/S8*(B(A,1)-B0)
1780 PRINT "Sforzo N      = ";N
1790 PRINT
1800 PRINT "X(C)          = ";X0
1810 PRINT "Y(C)          = ";Y0

```

Figura 4c


```

1820 PRINT
1830 PRINT "Errore sulla distanz
      a r di C dall'asse neutro n
      = ";E
1840 PRINT
1850 PRINT "Inclinazione di n su
      x = ";F
1860 PRINT
1870 PRINT "Distanza r di C da n
      = ";B0
1880 PRINT
1890 PRINT "I(n) = ";I0
1900 PRINT "S(n) = ";S0
1910 PRINT
1920 PRINT "Distanza da n del bo
      rdo compres-so = ";-(B0-T1)
1930 IF A=0 THEN 1960
1940 PRINT
1950 PRINT "Distanza da n del bo
      rdo teso = ";B(A,1)-B0
1960 PRINT
1970 PRINT
1980 PRINT "Sigma c = ";Z
1990 IF A=0 THEN 2010
2000 PRINT "Sigma f = ";Z1
2010 PRINT
2020 PRINT

```

Figura 4d

L'array V contiene le coordinate dei vertici:

$$V = \begin{vmatrix} x_1 & y_1 \\ x_2 & y_2 \\ \cdot & \cdot \\ \cdot & \cdot \\ \cdot & \cdot \\ x_i & y_i \end{vmatrix} ;$$

l'array A le coordinate dei punti ove sono concentrate le aree dell'acciaio, e le aree stesse *non* amplificate:

$$A = \begin{vmatrix} x_1 & y_1 & A_1 \\ x_2 & y_2 & A_2 \\ \cdot & \cdot & \cdot \\ \cdot & \cdot & \cdot \\ \cdot & \cdot & \cdot \\ x_j & y_j & A_j \end{vmatrix} ;$$

l'array D contiene le distanze dei vertici da a e b (3); l'array B serve per ordinare, alla fine del programma, le aree in ragione decrescente della distanza dall'asse neutro, onde ottenere la massima σ_f di trazione.

Nei DATA vanno forniti, nell'ordine per righe successive, prima gli elementi di V , e poi quelli di A ; si ricordi che le righe di V sono ordinate così come si susseguono i vertici nella già data numerazione. Prima ancora però devono essere forniti

N	sforzo normale, con il suo segno
NO	coefficiente di amplificazione
XO	ascissa x_0 di C
YO	ordinata y_0 di C
E	errore ammesso nel calcolo di b_0
$K1$	errore ammesso nel calcolo di φ
K	intervallo iniziale nella ricerca di due valori di φ tra cui c cambia di segno
F	valore iniziale di φ
H	base per il grafico
V	numero dei vertici + 1
A	numero delle aree.

I valori degli angoli sono forniti in gradi (frase 150). Le frasi 160 ÷ 180 e 240 ÷ 330 introducono i dati in memoria. Le frasi 340 ÷ 390 e 410 ÷ 530 forniscono i valori (1) e (2) delle caratteristiche d'inerzia delle aree metalliche rispetto agli assi xy ed ab . Le frasi 540 ÷ 570 introducono nell'array D le distanze dei vertici dalle rette a e b (3).

Le frasi 580 ÷ 650 calcolano (fig. 2) gli estremi $T1$ e $T2$ delle distanze dei vertici dall'asse a .

Il ciclo di tentativi relativo alla ricerca di C' una volta fissata la a' (fig. 3) è contenuto nel looping 780 ÷ 1060 (fig. 4). Si procede (fig. 5) in ogni loop al calcolo degli enti (4) per il trapezio generico della fig. 5, scartando (frase 790) i segmenti $V_i V_{i+1}$ paralleli all'asse a ed i trapezi (frase 800) tutti al di sopra di a' . Il trapezio si divide, a ciò fare, in strisce di altezza $p = \frac{d_{a(i+1)} - d_{ai}}{m}$ (frase 810). Le frasi da 840 ad 880 pongono

in $A6 \dots J6$ i valori già calcolati per le aree metalliche; ciò è fatto (frasi 770, 830 e 1050) soltanto nel primo loop. I valori b (B) e d (D) (fig. 5) sono calcolati nelle frasi 900 e 910.

Il test 920 ÷ 950 fa saltare in 1020 (e cioè esclude dalle sommatorie

970 ÷ 1010 i termini relativi alla striscia sotto esame), se $b > b_0$, ove il tratto $V_i V_{i+1}$ sia percorso in discesa. Questo test serve ad escludere il con-




Figura 5

tributo della parte di trapezio al disopra della retta a' . La frase 1040, esaurite tutte le strisce del trapezio, fa scendere in 1050, e quindi riporta in 780 (W resta = 1) al nuovo loop, e cioè al trapezio successivo.

Le frasi 1070 ÷ 1110 calcolano b_1 e b_3 . Inizialmente (frasi 710 e 720) B_4 e B_5 sono nulle, e quindi i test 1110 e 1130 fanno scendere in 1140 e 1150, assegnando a B_4 e B_5 i valori calcolati in b_3 e b_1 . Il numero TO aumenta di una unità (1160), e si ritorna (1170) in 730, e quindi (si

ha ancora $M = 10$) in 760; b_0 aumenta di un PO , e si ripete il procedimento. Questa volta però le frasi dei 1110 e 1130 non sono sicuramente false; se b_3 cambia di segno rispetto al suo precedente valore, e se contemporaneamente il vecchio valore di b_1 era negativo (vedi prob. 4) si passa al test di errore 1180 su E .

Se l'errore è contenuto nei limiti previsti, si passa in 1280, ottenendo (fig. 2) $c = CC'$ attraverso la (5); se non, si raddoppiano M ed MO e si ripete il procedimento. Alla fine (1280) c è contenuto in C ; il test 1290 chiude il calcolo se c è minore di $T2/1000$ (non si può pretendere che c sia nullo).

Le frasi da 1300 a 1460 ripetono il calcolo precedente per successivi valori di φ , incrementati di K , finché il valore di c cambia segno. Il ciclo si ripete per $K/10$, e così via, finché la differenza tra i due φ successivi per cui c cambia segno risulta minore di $K1$.

Esempio n. 1.

Si consideri il triangolo isoscele della fig. 6, di altezza $h = 6$ cm; il




Figura 6

centro di sollecitazione C sta sull'asse di simmetria, a distanza $h/4$ dalla base. Quindi C appartiene al contorno del nocciolo, e l'asse neutro n è la

```

10 DATA -1000,10,12,5.5
20 DATA .01,1,10
30 DATA 0,25,4,0
40 DATA 12,10,6,4,18,4,12,10
70 REM Programma "GLORIA"
    
```

Sforzo N = -1000

$X(C)$ = 12
 $Y(C)$ = 5.5

Errore sulla distanza r di C dal
 l'asse neutro n = .01

Inclinazione di n su x = 0

Distanza r di C da n = 4.5

$I(n)$ = 647.46
 $S(n)$ = -143.91

Distanza da n del bordo, compres-
 so = -6

Siema c = -41.6927246195




Figura 7

parallela ad x per il vertice 1. Risulta da ciò

$$r = 4,5 \text{ cm}$$

$$I_n = \frac{12 \cdot 6^3}{36} + \frac{12 \cdot 6}{2} \cdot 4^2 = 648 \text{ cm}^4$$




Figura 8

$$S_n = \frac{12 \cdot 6}{2} \cdot 4 = 144 \text{ cm}^3$$

e, per $N = -1000 \text{ Kg.}$,

$$\sigma_c = -\frac{1000}{144} \cdot 6 = 41,666 \text{ Kg cm}^{-2}$$

```

10 DATA -100,10,7,9,6,6
20 DATA 01,1,10
30 DATA 0,25,17,0
40 DATA 10,10,25,10,15,5,15,5,5
   5,5,5,5,5
50 DATA 5,5,9,75,10,9,75,10,10,
   25,5,5,10,25,5,5,14,5,9,5,14
   5,9,5,10,25,10,10,25
60 DATA 0,0,0
70 REM Programma "GLORIA"
    
```

Sforzo N = -100

X(C) = 7.9

Y(C) = 6.6

Errore sulla distanza r di C dal
l'asse neutro n = .01

Inclinazione di n su x = 19

Distanza r di C da n =
4.45329307166

I(n) = 139.090616176

S(n) = -31.617322217

Distanza da n del bordo compres-
so = -6.64981591698

Sigma c = -21.0321920096




Figura 9a

Errore C = .400111841834
 Incl.n su x = 0
 Dist.C da n = 4.65

Errore C = .200140306438
 Incl.n su x = 10
 Dist.C da n = 4.48943747318

Succes.val.F = 10
 Succes.val.K = 1

Errore C = .200140306438
 Incl.n su x = 10
 Dist.C da n = 4.48943747318

Errore C = .178494608372
 Incl.n su x = 11
 Dist.C da n = 4.49088770303

Errore C = .158043145679
 Incl.n su x = 12
 Dist.C da n = 4.49096996495

Errore C = .134665748025
 Incl.n su x = 13
 Dist.C da n = 4.48968423398

Errore C = .112640703978
 Incl.n su x = 14
 Dist.C da n = 4.48703090168

Errore C =
 8.57296250455E-2
 Incl.n su x = 15
 Dist.C da n = 4.48301077626

Errore C =
 6.48830134679E-2
 Incl.n su x = 16
 Dist.C da n = 4.47762508238

Errore C =
 4.23748860155E-2
 Incl.n su x = 17
 Dist.C da n = 4.47087546052

Errore C =
 2.02730786307E-2
 Incl.n su x = 18
 Dist.C da n = 4.46276396662

Figura 9b

Nella fig. 7 sono riportati i risultati del programma.

Esempio n. 2.

Nella fig. 8 è riportata la sezione di una spalla da ponte in conglomerato non armato, già studiata con metodo grafico nel cap. 7 del vol. 2 di "Fondamenti di Scienza delle Costruzioni"; nella fig. 9 sono riportati i risultati del programma.

Esempio n. 3.

Nella fig. 10 è disegnata una sezione in conglomerato armato, ripresa





Figura 10

da Giangreco (Tecnica delle Costruzioni, vol. 1, ed. 1975, pag. 142); i risultati che possono leggersi in questo testo, dopo il terzo calcolo grafico di approssimazione sono

```

10 DATA -26000,10,113,47.5
20 DATA .01,1,10
30 DATA -80,300,5,10
40 DATA 100,100,180,100,160,220
   120,220,100,100
50 DATA 103,103,3.14,118,4,103,
   3.14,133.8,103,3.14,149,2,10
   3,3.14,164.6,103,3.14,177,10
   3,3.14
60 DATA 123,217,7.68,134,3,217,
   7.68,145,66,217,7.68,157,217,
   7.68
70 REM Programma "GLORIA"

```



```

Storzo N      = -26000
X(C)          = 113
Y(C)          = 47.5
Errore sulla distanza r di C dal
l'asse neutro n = .01,
Inclinazione di n su x = -49.7
Distanza r di C da n =
76.5996386039
I(n) = 2130667.6375
S(n) = -27912.3421298
Distanza da n del bordo compres-
so = -52.4733417961
Distanza da n del bordo teso =
66.5886355301
Sigma c = -48.8782661217
Sigma f = 620.264869115

```

Figura 11a

Errore C = 55.31735188
 Incl.n su x = -80
 Dist.C da n = 36.2771517887

Errore C = 40.0034070409
 Incl.n su x = -70
 Dist.C da n = 54.4520406777

Errore C = 8.40551882819
 Incl.n su x = -60
 Dist.C da n = 65.374355653

Errore C = 6.58958079954
 Incl.n su x = -50
 Dist.C da n = 79.1815026356

Succes.val.F = -50
 Succes.val.K = 1

Errore C = 6.58958079954
 Incl.n su x = -50
 Dist.C da n = 79.1815026356

Succes.val.F = -50
 Succes.val.K = 1

Errore C = 6.58958079954
 Incl.n su x = -50
 Dist.C da n = 79.1815026356

Errore C = 6.58269026028
 Incl.n su x = -49.9
 Dist.C da n = 79.3080661823

Errore C = 6.57539654348
 Incl.n su x = -49.8
 Dist.C da n = 79.4343881435

Errore C = 6.56770134927
 Incl.n su x = -49.7
 Dist.C da n = 79.5604681337

Succes.val.F = -49.7
 Succes.val.K = .01


```

10 DATA -200000,10,162.5,50
20 DATA .001,1,10
30 DATA 0,400,13.2
40 DATA 225,253.5,100,253.5,100
   ,233.5,150,233.5,150,120,100
   ,120
50 DATA 100,100,225,100,225,120
   ,175,120,175,233.5,225,233.5
   ,225,253.5
60 DATA 162.5,250,36.94,162.5,1
   03.5,24.63
70 REM Programma "GLORIA"
    
```

Sforzo N = -200000

X(C) = 162.5

Y(C) = 50

Errore sulla distanza r di C dal
l'asse neutro n = 001

Inclinazione di n su x = 0

Distanza r di C da n =
99.8875

I(n) = 2515647.7531

S(n) = -25328 1070902

Distanza da n del bordo compres-
so = -49.9875

Distanza da n del bordo teso =
100.1125

Siema c = -116.930989568

Siema f = 2346.5304321




Figura 13

larghezza ali = 125
 larghezza anima = 25
 altezza totale = 153.5
 altezza ala sup. = 20
 altezza ala inf. = 20
 copriferro sup. = 3.5
 copriferro inf. = 3.5
 coeff. amplif. = 10
 errore radice r = 0.001
 area ferri sup. = 24.63
 area ferri inf. = 36.94
 sforzo normale = -200000

distanza dal bordo superiore del
 baricentro del solo conglomerato
 $r_0 = 76.75$

distanza dal bordo superiore del
 baricentro della sezione beton
 + ferro = 77.816705508

distanza dal bordo superiore del
 baricentro del solo ferro =
 91.3952411889

eccentricita' rispetto al bari-
 centro del solo conglomerato =
 -126.75

asse n nell'anima

quota asse neutro =
 49.8659515382

sigma c = -116.994454116
 sigma f sup. = -1087.82827209
 sigma f inf. = 2349.32413337




Figura 14

```

10 DATA -400,10,162.5,-10000
20 DATA .001,1,10
30 DATA 0,400,13,2
40 DATA 225,253.5,100,253.5,100
   ,233.5,150,233.5,150,120,100
   ,120.
50 DATA 100,100,225,100,225,120
   ,175,120,175,233.5,225,233.5
   ,225,253.5
60 DATA 162.5,250,36.94,162.5,1
   ,03.5,24.63
70 REM Programma "GLORIA"

```

```

Storzo N      = -400
X(C)          = 162.5
Y(C)          = -10000
Errore sulla distanza r di C dal
l'asse neutro n = 001
Inclinazione di n su x = 0
Distanza r di C da n =
10126.1429688
I(n) = 6529841
S(n) = -650.56118204
Distanza da n del bordo compres-
so = -26.1429688
Distanza da n del bordo teso =
123.8570312

```

```

Sigma c = -16.0741031108
Sigma f = 761.539634515

```




Figura 15

larghezza ala = 125
 larghezza anima = 25
 altezza utile = 150
 altezza ala = 20
 COPRI FERRO SUP = 3.5
 COEFF. AMPLIF. = 10
 area ferri SUP. = 24.63
 area ferri inf. = 36.94
 momento = 4000000
 raggio = 1000

quota asse neutro =
 25.942985996

momento d'inerzia =
 6529706.49596
 momento statico =
 45826.6609731

sigma c' = -15.8922830679
 sigma f' = -137.482357039
 sigma f = 759.954611013
 tau max = .280727233308




Figura 16




Figura 17

gono da un programma più semplice — e quindi più rapido — relativo alla pressoflessione delle travi a doppio T in c.a..

Nella fig. 15 sono poi riportati i valori forniti dal programma per $y_c = -10.000$ ed $N = -400 \text{ Kg}$, e cioè, in pratica, per un momento flettente in flessione semplice di $4.000.000 \text{ Kg cm}$, che comprime le fibre inferiori; nella fig. 16 si hanno i valori che si ottengono da un programma più semplice relativo alla flessione semplice delle travi a T in c.a..

Si fa osservare che proprio in questo caso, e cioè per grandi eccentricità, entra in gioco il test delle frasi 1190 — 1220. Ed infatti, nella fig. 17 è riportato il diagramma che lega la posizione dell'asse neutro a' e quella del centro C' . Quando l'asse a' contiene il bordo inferiore, C' è con buona approssimazione nel baricentro dell'armatura superiore (coinciderebbe con esso se il copriferro inferiore fosse nullo). Se poi l'asse a' si sposta verso l'alto, C' sale, fino a $+\infty$ quando a' coincide con l'asse neutro n_f della flessione semplice (con bordo inferiore, come già detto, compresso). E' ben evidente quindi che ad assi neutri compresi tra x ed n_f , e cioè a centri tra H e $+\infty$, corrisponde tenso flessione (sempre con bordo inferiore compresso), e cioè $N > 0$; e ad assi neutri compresi tra n_f e il bordo superiore, e cioè a centri compresi tra $-\infty$ e H' , corrisponde pressoflessione.

Se a' sale oltre n_f , C' passa da $+\infty$ a $-\infty$, e poi riprende a salire da $-\infty$. Assegnato perciò C , il programma trova i successivi C' per a' che sale da x , e fissa n , asse neutro di C , nella retta a' per cui $C' = C$ (questo per sezioni simmetriche, come nella fig. 17, se no la condizione è che $d_{C'x} = d_{Cx}$). Tale condizione si traduce nel cambiamento di segno di $B3$ per due posizioni successive di a' (frase 1120), e cioè per due successivi valori di BO ; per evitare che il test scatti quando a' supera n_f (fig. 17), la frase 1130 impone che, oltre alla condizione precedente (inversione di segno di $B3$), sia anche verificato che il precedente $B1$ sia negativo. Orbene, se C è molto giù rispetto alla sezione, accade che per NO non molto elevato (per esempio pari a 10, come fissato nel programma in prima approssimazione) e quindi per intervalli PO (frase 600) piuttosto grossi, il cambio di segno di $B3$ nel passaggio di a' attraverso l'asse neutro corrispondente a C non si avverta, e quindi il test 1120 non funziona. Ad evitare che così il programma vada in tilt valgono le istruzioni 1190-1220; esse impongono che, se BO non è minore di $T2$, se cioè l'asse a' non taglia la figura (e ciò accade appunto quando, per aver soltanto il test 1120, risulta $BO = T2$), si debba ripetere il procedimento con un passo PO ridotto alla metà; e così via.

2) Un programma per la verifica al taglio delle sezioni sottili monocollesse.

La linea media (fig. 18) sia costituita da tratti rettilinei, sia T il nu-

mero totale dei tratti, N quello dei tratti privi di estremi liberi (tratti *interni*). Il verso positivo di percorrenza è, per ogni tratto *esterno*, quello che inizia dall'estremo libero; per il generico tratto interno deve essere tale




Figura 18

che il suo estremo iniziale coincida con il finale di tutti gli altri tratti ad esso adiacenti; Si numerano i tratti, prima gli esterni, poi gli interni, per ciascuno di questi deve verificarsi che i tratti confluenti nel suo estremo iniziale abbiano numero d'ordine inferiore.

Nella fig. 19 è riportato il listing del programma, privo, per brevità, della parte relativa al grafico.

Nei DATA sono forniti prima T ed N ; poi, successivamente, seguendo l'ordinamento dei tratti, prima le coordinate $x_1 y_1 x_2 y_2$ dei due estremi di ogni tratto, e poi il suo spessore δ . La frase 680 legge T ed N ; le frasi 690 ÷ 730 pongono nelle prime cinque colonne dell'array A (che ha dimensioni $T \times 16$) le coordinate e gli spessori. La riga generica i -esima ($i = 1, 2$

```

10  REM Programma "TAGL.1"
640 OPTION BASE 1
650 DEG
660 DIM A(20,16)
670 INTEGER B(10,5)
680 READ T,N
690 FOR I=1 TO T
700 FOR J=1 TO 5
710 READ A(I,J)
720 NEXT J
730 NEXT I
740 FOR I=1 TO N
750 FOR J=1 TO 5
760 DISP "B(";I;",";J;")";
770 INPUT B(I,J)
780 NEXT J
790 NEXT I
800 DISP "Base H = ";
810 INPUT H
820 A1=0
830 S1=0
840 S2=0
850 I1=0
860 I2=0
870 J=0
880 FOR I=1 TO T
890 IF A(I,1)=A(I,3) THEN 920
900 A=ATN((A(I,2)-A(I,4))/(A(I,1)
    )-A(I,3))
910 GOTO 930
920 A=90
930 L=SQR((A(I,1)-A(I,3))^2+(A(I,2)-A(I,4))^2)
940 A(I,6)=L
950 A2=A(I,5)*L
960 A1=A1+A2
970 S1=S1+A2/2*(A(I,2)+A(I,4))
980 S2=S2-A2/2*(A(I,1)+A(I,3))
990 I8=A(I,5)*L^3/12
1000 I9=A(I,5)^3*L/12
1010 I1=I1+I9*COS(A)^2+I8*SIN(A)
    ^2+A2/4*(A(I,2)+A(I,4))^2
1020 I2=I2+I9*SIN(A)^2+I8*COS(A)
    ^2+A2/4*(A(I,1)+A(I,3))^2
1030 J=J+(I9-I8)*SIN(A)*COS(A)-A
    2/4*(A(I,1)+A(I,3))*(A(I,2)
    +A(I,4))
1040 NEXT I
1050 D1=S1/A1
1060 D2=S2/A1
1070 I3=I1-A1*D1^2
1080 I4=I2-A1*D2^2
1090 J1=J-A1*D1*D2
1100 IF I3#I4 THEN 1160
1110 F5=0
1120 F6=90
1130 I5=I3
1140 I6=I4
1150 GOTO 1200
1160 F5=ATN(2*J1/(I3-I4))/2
1170 F6=F5+90

```

Figura 19a

```

1180 I5=I4*SIN(F5)^2+I3*COS(F5)^
2+2*J1*SIN(F5)*COS(F5)
1190 I6=I4*SIN(F6)^2+I3*COS(F6)^
2+2*J1*SIN(F6)*COS(F6)
1200 R5=SQR(I5/A1)
1210 R6=SQR(I6/A1)
1220 PRINT "Area
          =" ; A1
1230 PRINT
1240 PRINT "Distanza del baricen
tro da x      =" ; D1
1250 PRINT
1260 PRINT "Distanza del baricen
tro da y      =" ; D2
1270 PRINT
1280 PRINT "Inclinazione dell'as
se 1          =" ; F5
1290 PRINT
1300 PRINT "Momento d'inerzia de
ll'asse 1     =" ; I5
1310 PRINT
1320 PRINT "Raggio sull'asse 1
          =" ; R6
1330 PRINT
1340 PRINT
1350 PRINT "Inclinazione dell'as
se 2          =" ; F6
1360 PRINT
1370 PRINT "Momento d'inerzia de
ll'asse 2     =" ; I6
1380 PRINT
1390 PRINT "Raggio sull'asse 2
          =" ; R5
1400 PRINT
1410 PRINT
1420 FOR I=1 TO T
1430 K=(A(I,4)-A(I,2))/A(I,6)
1440 A(I,7)=ACS((A(I,3)-A(I,1))/
A(I,6))
1450 IF K>=0 THEN 1470
1460 A(I,7)=-A(I,7)
1470 A(I,8)=(A(I,1)*A(I,4)-A(I,3
)*A(I,2))/A(I,6)
1480 NEXT I
1490 FOR I=1 TO T-N
1500 A(I,9)=0
1510 A(I,13)=0
1520 S3=A(I,5)*A(I,6)/2*(3*A(I,2
)+A(I,4))/4-A(I,5)*A(I,6)*D
1/2
1530 S4=-A(I,5)*A(I,6)/2*(3*A(I,
1)+A(I,3))/4-A(I,5)*A(I,6)*
D2/2
1540 A(I,10)=- (S4*SIN(F5)+S3*COS
(F5))/I5/A(I,5)
1550 A(I,14)=- (S4*SIN(F6)+S3*COS
(F6))/I6/A(I,5)
1560 S5=A(I,5)*A(I,6)*(A(I,2)+(A
(I,4)-A(I,2))/2)-A(I,5)*A(I
6)*D1

```

Figura 19b

```

1570 S6=-A(I,5)*A(I,6)*(A(I,1)+(
A(I,3)-A(I,1))/2)-A(I,5)*A(
I,6)*D2
1580 A(I,11)=- (S6*SIN(F5)+S5*COS
(F5))/I5/A(I,5)
1590 A(I,15)=- (S6*SIN(F6)+S5*COS
(F6))/I6/A(I,5)
1600 A(I,12)=(4*A(I,10)+A(I,11))
/6*A(I,5)*A(I,6)
1610 A(I,16)=(4*A(I,14)+A(I,15))
/6*A(I,5)*A(I,6)
1620 NEXT I
1630 FOR I=T-N+1 TO T
1640 Z1=0
1650 Z2=0
1660 FOR J=2 TO 5
1670 K=I-T+N
1680 IF B(K,J)=0 THEN 1710
1690 Z1=Z1-A(B(K,J),11)*A(B(K,J)
,5)*I5
1700 Z2=Z2-A(B(K,J),15)*A(B(K,J)
,5)*I6
1710 NEXT J
1720 A(I,9)=-Z1/A(I,5)/I5
1730 A(I,13)=-Z2/A(I,5)/I6
1740 S3=A(I,5)*A(I,6)/2*(3*A(I,2
)+A(I,4))/4-A(I,5)*A(I,6)*D
1/2
1750 S4=-A(I,5)*A(I,6)/2*(3*A(I,
1)+A(I,3))/4-A(I,5)*A(I,6)*
D2/2
1760 A(I,10)=- (Z1+S4*SIN(F5)+S3*
COS(F5))/I5/A(I,5)
1770 A(I,14)=- (Z2+S4*SIN(F6)+S3*
COS(F6))/I6/A(I,5)
1780 S5=A(I,5)*A(I,6)*(A(I,2)+(A
(I,4)-A(I,2))/2)-A(I,5)*A(I
,6)*D1
1790 S6=-A(I,5)*A(I,6)*(A(I,1)+(
A(I,3)-A(I,1))/2)-A(I,5)*A(
I,6)*D2
1800 A(I,11)=- (Z1+S6*SIN(F5)+S5*
COS(F5))/I5/A(I,5)
1810 A(I,15)=- (Z2+S6*SIN(F6)+S5*
COS(F6))/I6/A(I,5)
1820 A(I,12)=(A(I,9)+4*A(I,10)+A
(I,11))/6*A(I,5)*A(I,6)
1830 A(I,16)=(A(I,13)+4*A(I,14)+
A(I,15))/6*A(I,5)*A(I,6)
1840 NEXT I
1850 X1=0
1860 Y1=0
1870 Z1=0
1880 X2=0
1890 Y2=0
1900 Z2=0
1910 FOR I=1 TO T
1920 X1=X1+A(I,12)*COS(A(I,7))
1930 Y1=Y1+A(I,12)*SIN(A(I,7))
1940 Z1=Z1+A(I,12)*A(I,8)

```

Figura 19c

```

1950 X2=X2+A(I,16)*COS(A(I,7))
1960 Y2=Y2+A(I,16)*SIN(A(I,7))
1970 Z2=Z2+A(I,16)*A(I,8)
1980 NEXT I
1990 X0=(X1*Z2-X2*Z1)/(X1*Y2-X2*
    Y1)
2000 Y0=(Y1*Z2-Y2*Z1)/(X1*Y2-X2*
    Y1)
2010 PRINT "Asse neutro = asse 1
"
2020 PRINT
2030 PRINT "Il taglio T=1 e'dire
    tto secondo l'asse 2 , e' la
    coppia 1T e'antioraria."
2040 PRINT
2050 FOR I=1 TO T
2060 PRINT "Tau(";I;",".1) = ";A(I
    ,9)
2070 PRINT "Tau(";I;",".m) = ";A(I
    ,10)
2080 PRINT "Tau(";I;",".2) = ";A(I
    ,11)
2090 PRINT
2100 NEXT I
2110 PRINT
2120 PRINT
2130 PRINT "Asse neutro = asse 2
"
2140 PRINT
2150 PRINT "Il taglio T=1 e'dire
    tto secondo l'asse 1 , e' la
    coppia 2T e'antioraria "
2160 PRINT
2170 FOR I=1 TO T
2180 PRINT "Tau(";I;",".1) = ";A(I
    ,13)
2190 PRINT "Tau(";I;",".m) = ";A(I
    ,14)
2200 PRINT "Tau(";I;",".2) = ";A(I
    ,15)
2210 PRINT
2220 NEXT I
2230 PRINT
2240 PRINT
2250 PRINT "Ascissa x del centro
    di taglio = ";X0
2260 PRINT
2270 PRINT "Ordinata y del centr
    o di taglio =" ;Y0
2280 PRINT
2290 PRINT

```

Figura 19d

... T) dell'array A si presenta come segue (è spezzata in due tratti per necessità grafiche)

1	2	3	4	5	6	7	8
x_{i1}	y_{i1}	x_{i2}	y_{i2}	δ_i	l_i	α_i	d_i
9	10	11	12	13	14	15	16
τ'_{i1}	τ'_{im}	τ'_{i2}	F'_i	τ''_{i1}	τ''_{im}	τ''_{i2}	F''_i ;

l_i (fig. 20) è la lunghezza; α_i l'angolo di inclinazione sull'asse x ; d_i la distanza dall'origine O ($d_i > 0$ se O è alla sinistra di t); τ_1 τ_m τ_2 i valori della τ in 1, m e 2, ove m è il punto medio del tratto, F_i il valore della risultante

$\delta \int_1^2 \tau ds$. Gli apici semplice e doppio si riferiscono il primo alla forza




Figura 20

tagliante unitaria agente secondo l'asse principale 1, e il secondo alla forza tagliante unitaria agente secondo l'asse principale 2. Le frasi da 820 a 1410 sono relative alla ricerca delle caratteristiche d'inerzia della

sezione (vedi 1° vol. di Problemi); in più, la frase 940 inserisce le l_i nella sesta colonna dell'array A .

Le frasi da 1430 a 1460 calcolano α_i . A tale proposito si osservi che già α_i era fornito dalla frase 900, come arco tangente, e quindi compreso fra $+90^\circ$ e -90° , ed a meno di 180° ; occorre adesso, invece, conoscere l'effettivo valore di α_i , fra $+180^\circ$ e -180° , in relazione al verso 1-2 della retta t . Si calcola perciò

$$K = \text{sen } A(I, 7) = \frac{y_2 - y_1}{l}$$

$$A(I, 7) = \text{arc cos } \frac{x_2 - x_1}{l}; \quad (6)$$

$A(I, 7)$ è compreso fra 0 e 180° , ma la (6) è soddisfatta anche dal suo simmetrico. Scioglie il dubbio il segno di K ; se infatti $K > 0$, α è compreso fra 0 e 180 , se $K < 0$, è invece compreso fra -180 e 0 .

La frase 1470 calcola poi d_i ; se infatti l'equazione di t è

$$ax + by + c = 0, \quad (7)$$

risulta (fig. 20)

$$OA = -\frac{c}{b},$$

e quindi

$$d = \frac{c}{b} \cos \alpha. \quad (8)$$

La (8) fornisce d con il suo segno; se il verso di t cambia, cambia anche il segno di $\cos \alpha$.

Per esplicitare le costanti della (7) si osservi che l'equazione di t è

$$\frac{x_1 - x}{y_1 - y} = \frac{x_1 - x_2}{y_1 - y_2}$$

e quindi

$$c = x_2 y_1 - x_1 y_2$$

$$b = x_1 - x_2 .$$

La (8) perciò può scriversi

$$d = \frac{x_2 y_1 - x_1 y_2}{x_2 - x_1} \cos \alpha$$

e ancora

$$d = \frac{x_2 y_1 - x_1 y_2}{l} ; \quad (9)$$

infatti $\cos \alpha$ cambia di segno con $x_2 - x_1$.

Si formano così, con il looping $1410 \div 1480$, la sesta e settima colonna dell'array A .

Dal display sono richiesti i dati dell'array B (frasi $740 \div 790$). Gli elementi di questo array sono numeri interi, esso è stato perciò dimensionato (frase 670) con INTEGER invece che con DIM, per risparmio di memoria operativa. L'array B per la sezione della fig. 18 è il seguente:

$$B = \begin{vmatrix} 7 & 1 & 0 \\ 8 & 7 & 2 \\ 9 & 8 & 0 \\ 10 & 3 & 4 \\ 11 & 6 & 5 \end{vmatrix} \quad (10)$$

Nella prima colonna sono disposti i numeri d'ordine degli N tratti interni; in ogni riga, al numero d'ordine del tratto seguono, in un ordine qualsiasi, quelli dei tratti che confluiscono nel suo estremo iniziale. La prima dimensione di B è N , la seconda è $M + 1$, ove M è il massimo numero di tratti confluenti in ogni estremo iniziale. Poichè in programma A è stato dimensionato come 20×16 , e B come 10×5 , è necessario modificare tali dimensioni, e cioè le frasi 660 e 670, solo se $T > 20$, $N > 10$, $M > 4$.

Il looping $1490 \div 1620$ è relativo ai tratti esterni, e quindi alle prime $T - N$ righe di A ; le τ_1 sono comunque nulle (frasi 1500 e 1510).

Si ha poi (fig. 21), per il punto medio del tratto,

$$(S3) \quad S_{x_0} = \frac{\delta l}{2} \left(y_1 + \frac{y_2 - y_1}{4} \right) - \frac{\delta l}{2} d_{G_x} =$$

$$= \frac{\delta l}{2} \cdot \frac{3y_1 + y_2}{4} - \frac{\delta l}{2} d_{Gx}$$

$$(S4) \quad S_{y_0} = -\frac{\delta l}{2} \cdot \frac{3x_1 + x_2}{4} - \frac{\delta l}{2} d_{Gy} .$$

Si ha poi




Figura 21

$$S_{\xi} = S_{y_0} \operatorname{sen} \varphi + S_{x_0} \operatorname{cos} \varphi$$

$$S_{\eta} = S_{y_0} \operatorname{sen} (\varphi + 90^{\circ}) + S_{x_0} \operatorname{cos} (\varphi + 90^{\circ}) ,$$

(11)

e quindi (frasi 1540 e 1550)

$$\tau'_m = - S_{\xi} / I_{\xi} \delta$$

$$\tau''_m = - S_{\eta} / I_{\eta} \delta .$$

Così pure, per il punto 2 del tratto si ha

$$(S5) \quad S_{x_0} = \delta l \left(y_1 + \frac{y_2 - y_1}{2} \right) - \delta l d_{Gx}$$

$$(S6) \quad S_{y_0} = -\delta l \left(x_1 + \frac{x_2 - x_1}{2} \right) - \delta l d_{Gy} ,$$

e quindi si ottengono τ'_2 e τ''_2 (frasi 1580 e 1590) con formule analoghe alle (11) e (12).

Le frasi 1600 e 1610 forniscono poi le risultanti. Sono così completate le prime $T - N$ righe di A .

Il looping 1630 ÷ 1840 è analogo al precedente, e vale per i tratti interni. Il looping interno 1660 ÷ 1710, per il tratto definito dal valore I del looping esterno, esplora la corrispondente riga $(i - t + n)$ -esima dell'array B , e se essa trova un valore diverso da zero, questo è indice di un tratto che confluisce a monte del tratto in esame; quindi accumula in $Z1$ e $Z2$ i corrispondenti valori

$$- \tau'_2 \delta I_\xi$$

$$- \tau''_2 \delta I_\eta.$$

Le frasi 1720 e 1730 calcolano quindi, per il tratto sotto I ,

$$\tau'_1 = \frac{1}{\delta} \sum \tau'_{i2} \delta_i$$

$$\tau''_1 = \frac{1}{\delta} \sum \tau''_{i2} \delta_i ,$$

dove δ è relativo al tratto sotto I , e le sommatorie sono estese ai tratti a monte.

La frase 1760 calcola

$$\tau'_m = \frac{\tau'_1 \delta I_\xi - S_\xi}{I_\xi \delta} = \tau'_1 - \frac{S_\xi}{I_\xi \delta} ;$$

il resto del looping è di facile comprensione.

Le frasi da 1850 a 1980 calcolano in X_1, Y_1, Z_1 ed in X_2, Y_2, Z_2 le due componenti secondo x ed y ed il momento rispetto ad O delle $\tau' dA$ e $\tau'' dA$.

Dalla fig. 22 si trae

$$Ya = Z$$

$$- Xb = Z$$

da cui




Figura 22

$$\begin{aligned} a &= Z/Y \\ &= - Z/X . \end{aligned} \tag{13}$$

L'equazione della retta d'azione di F sia

$$Ax + By + C = 0 ; \quad (14)$$

$$a = - C/A = Z/Y$$

$$b = - C/B = - Z/X .$$

La (14) pertanto si scrive

$$- Yx + Xy + Z = 0 . \quad (15)$$

La (15), ricavata per X, Y, Z non nulli, è però valida comunque. Il centro di taglio è l'intersezione delle due rette (15) scritte per X_1, Y_1, Z_1 e per X_2, Y_2, Z_2 ; quindi

$$x_C = \frac{X_1 \cdot Z_2 - X_2 \cdot Z_1}{X_1 \cdot Y_2 - X_2 \cdot Y_1} \quad (16)$$

$$y_C = \frac{Y_1 \cdot Z_2 - Y_2 \cdot Z_1}{X_1 \cdot Y_2 - X_2 \cdot Y_1} .$$

Le 1990 e 2000 calcolano le (16).

Esempio n. 1.

Nella trave a doppio T della fig. 23 le prime 5 colonne di A sono

$$A' = \begin{vmatrix} 8 & 16 & 12 & 16 & 0,4 \\ 16 & 16 & 12 & 16 & 0,4 \\ 8 & 7 & 12 & 7 & 0,4 \\ 16 & 7 & 12 & 7 & 0,4 \\ 12 & 16 & 12 & 7 & 0,2 \end{vmatrix}$$

E' poi




Figura 23

$$B = | 5 \quad 1 \quad 2 \quad 0 \quad 0 | .$$

I risultati del programma sono quelli della fig. 24:

```

2600 DATA 5,1
2610 DATA 8,16,12,16,4
2620 DATA 16,16,12,16,4
2630 DATA 8,7,12,7,4
2640 DATA 16,7,12,7,4
2650 DATA 12,16,12,7,2
    
```

Area
= 8.2

Distanza del baricentro da x
= 11.5

Distanza del baricentro da y
= -12

Inclinazione dell'asse 1
= 0


Momento d'inerzia dell'asse 1
= 141 83533333

Raggio sull'asse 1
= 2.04042479227

Inclinazione dell'asse 2
= 90

Momento d'inerzia dell'asse 2
= 34 13933333

Raggio sull'asse 2
= 4 15996524024



Ascissa x del centro di taglio =
12

Ordinata y del centro di taglio
= 11.5

Figura 24a